

統計函數(一)

筆數

- COUNT(範圍1,範圍2,...)
- COUNTA(範圍1,範圍2,...)
- 用來求算儲存格個數，於資料表中可用來計算記錄筆數。範圍引數，最多可達255個。
- COUNT()係計算含數值資料的儲存格個數；
- 而COUNTA()則計算非空白的儲存格個數

C11		fx =COUNTA(C2:C8)					
	A	B	C	D	E	F	G
1	學號	姓名	成績				
2	91001	李碧華	88				
3	91002	林淑芬	90				
4	91003	王嘉育	缺考				
5	91004	吳育仁	88				
6	91005	呂姿澄	75				
7	91006	孫國華	85				
8	91007	李慶昭	68				
9							
10	針對成績欄求筆數						
11	以COUNTA求		7	← =COUNTA(C2:C8)			
12	以COUNT求		6	← =COUNT(C2:C8)			
13							

筆數

- 兩函數之處理範圍均為C2:C8：
 - =COUNTA(C2:C8) =>7**
 - =COUNT(C2:C8) =>6**
- 因C4為“缺考”字串並非數值，故COUNT()函數所計算之結果只為6筆
- 〔 B 欄 〕 只能以COUNTA()計算
- 若使用COUNT()其結果將為0
- 因為姓名欄是字串，且無數值資料型別

C11		fx =COUNTA(C2:C8)					
	A	B	C	D	E	F	G
1	學號	姓名	成績				
2	91001	李碧華	88				
3	91002	林淑芬	90				
4	91003	王嘉育	缺考				
5	91004	吳育仁	88				
6	91005	呂姿滢	75				
7	91006	孫國華	85				
8	91007	李慶昭	68				
9							
10	針對成績欄求筆數						
11	以COUNTA求		7	← =COUNTA(C2:C8)			
12	以COUNT求		6	← =COUNT(C2:C8)			
13							

C16		fx =COUNTA(B2:B8)					
	A	B	C	D	E	F	G
15	針對姓名欄求筆數						
16	以COUNTA求		7	← =COUNTA(B2:B8)			
17	以COUNT求		0	← =COUNT(B2:B8)			
18							

依條件算筆數

- COUNTIF(範圍,條件準則)
- 於指定之範圍內，依條件準則進行求算符合條件之筆數
- 條件準則可以是數字、比較式或文字
- 但除非使用數值，否則應以雙引號將其包圍。如：**70**、"男"或"**>=80**"

- 擬分別求男女人數
- G2求男性人數之公式：
=COUNTIF(C2:C8,"男")
 - 在C2:C8性別欄中，求算內容為"男"之人數
- G3求女性人數之公式：
=COUNTIF(C2:C8,"女")

[illegible]

依條件算筆數

- 特殊：比較符號之條件式一前後須加上雙引號 “”
- G5求成績80分以上之人數的公式：
 - =COUNTIF(D2:D8, ">=80")
 - 表要在D2:D8之成績欄中，求算內容大於等於80（">=80"）之人數。
- G6求80分以下之人數的公式：
 - =COUNTIF(D2:D8, "<80")
 - 求全班及格與不及格之人數

G2		=COUNTIF(C2:C8,"男")								
	A	B	C	D	E	F	G	H	I	J
1	學號	姓名	性別	成績						
2	91001	李碧華	女	70		男性人數	3	← =COUNTIF(C2:C8,"男")		
3	91002	林淑芬	女	89		女性人數	4	← =COUNTIF(C2:C8,"女")		
4	91003	王嘉育	男	78						
5	91004	吳育仁	男	82		80分及以上	4	← =COUNTIF(D2:D8,">=80")		
6	91005	呂姿澄	女	83		80分以下	3	← =COUNTIF(D2:D8,"<80")		
7	91006	孫國華	女	87						
8	91007	李慶昭	男	68						
9										

	A	B	C	D	E	F	G
1	學號	成績					
2	9079001	85					
3	9079002	25			及格	8	
4	9079003	60			不及格	6	
5	9079004	80					
6	9079005	87					
7	9079006	90					
8	9079007	28					
9	9079008	32					
10	9079009	80					
11	9079010	45					
12	9079011	85					
13	9079012	52					
14	9079013	88					
15	9079014	38					
16	9079015	68					
17							

計算空白儲存格

- COUNTBLANK(範圍)
- 計算指定範圍內，空白儲存格之個數。
- 如果格內之公式結果為虛字串（""），此儲存格仍會被計算在內；但若為0，則不予計算
- B10之公式內容為：
 - =COUNTBLANK(D2:D8)
 - 算出D2:D8內有兩個儲存格為空白

B10		fx =COUNTBLANK(D2:D8)					
	A	B	C	D	E	F	G
1	學號	姓名	性別	成績			
2	91001	李碧華	女	70			
3	91002	林淑芬	女	89			
4	91003	王嘉育	男				
5	91004	吳育仁	男	82			
6	91005	呂姿滢	女				
7	91006	孫國華	女	87			
8	91007	李慶昭	男	68			
9							
10	無成績者	2	人				
11							

依條件算加總

- SUMIF(準則範圍,條件準則,加總範圍)
- 求某欄中符合條件部份之加總
- 準則範圍是條件準則用來進行條件比較的範圍
- 條件準則可以是數字、比較式或文字。但除非使用數值，否則應以雙引號將其包圍。如：50000、"門市"或">=800000"

依條件算加總

- 加總範圍則用以標出要進行加總的儲存格範圍，如果省略，則計算準則範圍中的儲存格。僅適用於準則範圍為數值時，如：

- =SUMIF(C2:C8,">=30000")
- 將加總C2:C8範圍內，大於或等於30000者

- 求各部門之業績總和：

- G2求『門市』部之業績合計的公式：
- **=SUMIF(A2:A8,"門市",C2:C8)**
- 在A2:A8之部門欄中，求算內容為"門市"之業績合計。

[illegible]

依條件算加總

- G3求『業務』部之業績合計公式：
=SUMIF(A2:A8,"業務",C2:C8)
- 將業績分成三萬及以上與三萬以下兩組，並分別求其業績總和之公式：
 - =SUMIF(C2:C8,">=30000")
=SUMIF(C2:C8,"<30000")
或
=SUMIF(C2:C8,">=30000",C2:C8)
=SUMIF(C2:C8,"<30000",C2:C8)
- 因為，省略加總範圍將計算準則範圍中的儲存格（C2:C8）內容。

[illegible]

- 計算方式：
- AVERAGEIF()
- 將SUMIF()除以COUNTIF()
- 【分組均數】工作表

[illegible]

均數

- C11與C12處，同樣以C2:C8為處理範圍
=AVERAGEA(C2:C8)
=AVERAGE (C2:C8)
- 所求之均數卻不同？
- 因C4為"缺考"字串並非數值，故AVERAGE()會將其排除掉
- 也就是說其分母為6；而非AVERAGEA()的7。

C11			fx =AVERAGEA(C2:C8)			
	A	B	C	D	E	F
1	學號	姓名	成績			
2	91001	李碧華	88			
3	91002	林淑芬	90			
4	91003	王嘉育	缺考			
5	91004	吳育仁	88			
6	91005	呂姿潒	75			
7	91006	孫國華	85			
8	91007	李慶昭	68			
9						
10	求成績欄均數					
11	以AVERAGEA求		70.6	← =AVERAGEA(C2:C8)		
12	以AVERAGE求		82.3	← =AVERAGE(C2:C8)		
13						

均數

- 求學生平時作業之均數，第一位學生並未繳交『作業2』。
- 但無論以AVERAGE()或AVERAGEA()函數求算，均是只交兩次作業者的平均（85.0）高過三次全交者（79.3）。
- 應記得於未繳作業處輸入0。以避免前面之不合理情況
- 【成績】工作表

F2		fx =AVERAGE(C2:E2)							
	A	B	C	D	E	F	G	H	I
1	學號	姓名	作業1	作業2	作業3	平均			
2	91001	李碧華	88		82	85.0	← =AVERAGE(C2:E2)		
3	91002	林淑芬	90	60	88	79.3	← =AVERAGE(C3:E3)		
4									

F2		fx =AVERAGE(C2:E2)							
	A	B	C	D	E	F	G	H	I
1	學號	姓名	作業1	作業2	作業3	平均			
2	91001	李碧華	88	0	82	56.7	← =AVERAGE(C2:E2)		
3	91002	林淑芬	90	60	88	79.3	← =AVERAGE(C3:E3)		
4									

極大

- MAX(數值1,數值2, ...)
- MAXA(數值1,數值2, ...)
- 求算一串數值的極大值，範圍引數最多可達355個。
- MAX()求所有數值資料的極大值
- MAXA()求所有非空白儲存格之極大值

C11		fx =MAXA(C2:C8)				
	A	B	C	D	E	F
1	學號	姓名	成績			
2	91001	李碧華	88			
3	91002	林淑芬	90			
4	91003	王嘉育	缺考			
5	91004	吳育仁	88			
6	91005	呂姿潁	75			
7	91006	孫國華	85			
8	91007	李慶昭	68			
9						
10	求成績欄之極大值					
11	以MAXA求		90	← =MAXA(C2:C8)		
12	以MAX求		90	← =MAX(C2:C8)		
13						

極小

- MIN(數值1,數值2, ...)
- MINA(數值1,數值2, ...)
- 求算一串數值的極小值，範圍引數最多可達255個。
- MIN()求所有數值資料的極小值
- MINA()求所有非空白儲存格之極小值

C11			fx =MINA(C2:C8)			
	A	B	C	D	E	F
1	學號	姓名	成績			
2	91001	李碧華	88			
3	91002	林淑芬	90			
4	91003	王嘉育	缺考			
5	91004	吳育仁	88			
6	91005	呂姿潒	75			
7	91006	孫國華	85			
8	91007	李慶昭	68			
9						
10	求成績欄之極小值					
11	以MINA求		0	← =MINA(C2:C8)		
12	以MIN求		68	← =MIN(C2:C8)		
13						

生產之實例

- 某產品，須甲零件**10**、乙零件**3**、丙零件**4**，才可組合一個成品。
- 今庫房內分別有甲零件**170**、乙零件**60**、丙零件**85**。最多可組成幾個成品？
- 雖然，個別零件分別可完成**17**、**53**與**21**件成品，但得三者均有，才可組合成一完整之成品。
- 故其答案**17**
- 為求此三者的極小值
=MIN(D17:D19)

D21		=MIN(D17:D19)				
	A	B	C	D	E	F
16		零件數	完成一件要幾個零件	可完成數		
17	甲	170	10	17	<-- =INT(B17/C17)	
18	乙	160	3	53	<-- =INT(B18/C18)	
19	丙	85	4	21	<-- =INT(B19/C19)	
20						
21			這些零件最多可完成	17		
22						

體操評分

- 體操選手的比賽成績，為求公平（避免偏袒或惡意）會將最大與最小值先排除掉，再求其總分。
- 求個選手之總分數
- 公式：SUM(B2:H2)-MIN(B2:H2)-MAX(B2:H2)

[illegible]

母體標準差

- STDEVP(數值1,數值2,...)
- STDEVPA(數值1,數值2,...)
- 計算母體標準，數值1,數值2,...
- 為要計算標準差之儲存格或範圍引數，最多可達255個。
- STDEVP()係求所有數值的母體標準差
- STDEVPA()則求所有非空白儲存格之母體標準差

母體標準差

- 母體標準差的計算公式為

$$\sqrt{\frac{n \sum x^2 - (\sum x)^2}{n^2}}$$

- 其與樣本標準差之公式，只差在後者之分母為 $n*(n-1)$ ；而前者為 $n*n$ 。當樣本個數 n 愈大時，樣本標準差與母體標準差會愈趨近於相等

$$\sqrt{\frac{n \sum x^2 - (\sum x)^2}{n(n-1)}}$$

樣本變異數

- VAR(數值1,數值2,...)
- VARA(數值1,數值2,...)
- 計算樣本變異數，數值1,數值2,...
- 要計算變異數之儲存格或範圍引數，它是對應於某母群體抽樣選出的1到255個數值樣本。

- 樣本變異數的計算公式為：

$$\frac{n \sum x^2 - (\sum x)^2}{n(n-1)}$$

- 其值恰為樣本標準差之平方，也是用來衡量觀測值與平均值間的離散程度

樣本變異數

- **VAR()**係求所有數值的樣本變異數
- **VARA()**則求所有非空白儲存格之樣本變異數
- **E12**之公式，因將“缺考”，當成0納入計算，故其變異數明顯增大

E11		fx =VAR(C2:C8)					
	A	B	C	D	E	F	G
1	學號	姓名	成績				
2	91001	李碧華	88				
3	91002	林淑芬	90				
4	91003	王嘉育	缺考				
5	91004	吳育仁	88				
6	91005	呂姿澄	75				
7	91006	孫國華	85				
8	91007	李慶昭	68				
9							
10	標準差			變異數			
11	以STDEV	8.824209		以VAR	77.86667	← =VAR(C2:C8)	
12	以STDEVA	32.14476		以VARA	1033.286	← =VARA(C2:C8)	
13							

母體變異數

- **VARP(數值1,數值2,...)**
- **VARPA(數值1,數值2,...)**
- 計算母體變異數，數值1,數值2,...
- 要計算變異數之儲存格或範圍引數，它是對應於母群體的1到255個數字引數
- **VARP()**係求所有數值的母體變異數
- **VARPA()**則求所有非空白儲存格之母體變異數

母體變異數

- 母體變異數的計算公式為

$$\frac{n \sum x^2 - (\sum x)^2}{n^2}$$

- 其與樣本變異數之公式

$$\frac{n \sum x^2 - (\sum x)^2}{n(n-1)}$$

- 只差在後者之分母為 $n*(n-1)$ ；而前者為 $n*n$ 。當樣本個數 n 愈大時，樣本變異數與母體變異數會愈趨近於相等。

排名次

- RANK(數值,範圍,順序)
- 數值為要安排等級之數字（如：某人成績）
- 範圍是標定要將進行排名次之數值範圍（如：全班之成績），非數值將被忽略。
- 順序是用來指定排等級順序之方式
 - 為0或省略，表要遞減排等級，即數值大者在前；小者在後。
 - 若不是0，則表要遞增排等級，即數值小者在前；大者在後。
- 當有同值之情況，會給相同之等級。如，第三名有兩位同分，其等級均為3；且下一位就變成第5名，而無第4名。

排名次

- 成績資料排名次
- 其內有兩位學生之成績同為88分：
 - 於D2輸入= RANK(C2,\$C\$2:\$C\$8)表要根據C2之分數
 - 於\$C\$2:\$C\$8之全體成績內，以遞減方式排列名次（高分者在前）
 - 續將D2抄給D3:D8，即可排出正確之名次
- 可發現兩位同為88分之學生均為第2名，接下來之85分的學生即排為第4名（而非第3名）

	D2		fx		
	A	B	C	D	E
1	學號	姓名	成績	名次	
2	91001	李碧華	88		
3	91002	林淑芬	90		
4	91003	王嘉育	85		
5	91004	吳育仁	88		
6	91005	呂姿潒	75		
7	91006	孫國華	85		
8	91007	李慶昭	68		
9					

	D2		fx =RANK(C2,\$C\$2:\$C\$8)				
	A	B	C	D	E	F	G
1	學號	姓名	成績	名次			
2	91001	李碧華	88	2	←	=RANK(C2,\$C\$2:\$C\$8)	
3	91002	林淑芬	90	1	←	=RANK(C3,\$C\$2:\$C\$8)	
4	91003	王嘉育	85	4	←	=RANK(C4,\$C\$2:\$C\$8)	
5	91004	吳育仁	88	2	←	=RANK(C5,\$C\$2:\$C\$8)	
6	91005	呂姿潒	75	6	←	=RANK(C6,\$C\$2:\$C\$8)	
7	91006	孫國華	85	4	←	=RANK(C7,\$C\$2:\$C\$8)	
8	91007	李慶昭	68	7	←	=RANK(C7,\$C\$2:\$C\$8)	
9							

排名次

- 求體操選手之總分的名次，總分部份得排除最高及最低分

[illegible]

次數分配

- **FREQUENCY(資料陣列,組界範圍陣列)**
- 計算某一個範圍內各不同值出現的次數，但其回應值為一縱向之陣列，故輸入前應先選取相當陣列元素之儲存格，輸入公式後，以**Ctrl+Shift+Enter**完成輸入。
- 資料陣列是一個要計算次數分配的**數值陣列或數值參照位址**
- 組界範圍陣列是一個陣列或儲存格範圍參照位址，用來安排各答案之分組結果。

次數分配

- 假定，於問卷調查結果中，要求不同性別之人數的次數分配表。
- 輸入所有可能出現之答案
- 如A15:A16之1、2，作為組界範圍陣列
- 選取恰與答案數同格數之垂直範圍A15:A16
- 輸入
 - =FREQUENCY(B2:B12,A15:A16)

	B15		fx		
	A	B	C	D	E
1	問卷編號	性別	第一題	第二題	
2	1001	1	3	1	
3	1002	2	2	2	
4	1003	1	1	4	
5	1004	2	2	3	
6	1005	1	3	2	
7	1006	1	2	3	
8	1007	1	3	4	
9	1008	2	3	3	
10	1009	2	3	4	
11	1010	1	2	4	
12	1011	2	3	2	
13					
14	性別	人數	%		
15		1			
16		2			
17	總計				
18					

	PMT		=FREQUENCY(B2:B12,A15:A16)					
	A	B	C	D	E	F	G	H
1	問卷編號	性別	第一題	第二題				
2	1001	1	3	1				
3	1002	2	2	2				
4	1003	1	1	4				
5	1004	2	2	3				
6	1005	1	3	2				
7	1006	1	2	3				
8	1007	1	3	4				
9	1008	2	3	3				
10	1009	2	3	4				
11	1010	1	2	4				
12	1011	2	3	2				
13								
14	性別	人數	%					
15		1						
16		2						
17	總計							
18								

次數分配

- 按**Ctrl+Shift+Enter**完成輸入，即可完成一陣列之內容，求得各答案之次數分配表
- 於B17求其次數分配之總計；
- 於C15:C17，求其百分比，即可作成一完整之次數分配表
- 原公式左右以一對大括號（{ }）包圍，表其為陣列內容。
- 這兩格內容將視為一個整體，要刪除時必須兩個一起刪。
- 也無法僅單獨變更某一格之內容

	C15		fx =B15/\$B\$17			
	A	B	C	D	E	F
14	性別	人數	%			
15	1	6	54.5%			
16	2	5	45.5%			
17	總計	11	100.0%			
18						

	B15		fx {=FREQUENCY(B2:B12,A15:A16)}					
	A	B	C	D	E	F	G	H
1	問卷編號	性別	第一題	第二題				
2	1001	1	3	1				
3	1002	2	2	2				
4	1003	1	1	4				
5	1004	2	2	3				
6	1005	1	3	2				
7	1006	1	2	3				
8	1007	1	3	4				
9	1008	2	3	3				
10	1009	2	3	4				
11	1010	1	2	4				
12	1011	2	3	2				
13								
14	性別	人數	%					
15	1	6						
16	2	5						
17	總計							
18								

次數分配

- 問卷調查資料，求第一題及第二題之答案分佈情況

[illegible]

次數分配

- 想於各題下以一對一之方式，標示出其次數分配
- 以單一答案範圍
 - =A15:A18
 - =>作為組界範圍陣列
 - 供所有FREQUENCY()函數使用

D15		fx {=FREQUENCY(D2:D12,A15:A18)}						
	A	B	C	D	E	F	G	H
1	問卷編號	性別	第一題	第二題				
2	1001	1	3	1				
3	1002	2	2	2				
4	1003	1	1	4				
5	1004	2	2	3				
6	1005	1	3	2				
7	1006	1	2	3				
8	1007	1	3	4				
9	1008	2	3	3				
10	1009	2	3	4				
11	1010	1	2	4				
12	1011	2	3	2				
13								
14	答案	性別	第一題	第二題				
15	1	6	1	1				
16	2	5	4	3				
17	3		6	3				
18	4			4				
19								

分組資料之次數分配

- 將所得資料分成幾個區間，再計算落於各區間之所得分佈情況。
- H3:H6之組界範圍陣列，將其分為四個組別：
 - ~30000
 - 30000~50000
 - 50000~70000
 - 70000~
- 還可於G3:G6輸入字串，讓G3:H6看似標示區間之內容，更能讓使用者看出其次數分配結果所代表之意義
- 【[所得分組](#)】工作表

I3		={FREQUENCY(E2:E12,H3:H6)}							
	A	B	C	D	E	F	G	H	I
1	問卷編號	性別	第一題	第二題	所得				
2	1001	1	3	1	36000			所得分組	次數
3	1002	2	2	2	52000			30000	3
4	1003	1	1	4	64000			50000	3
5	1004	2	2	3	18000			70000	3
6	1005	1	3	2	22000				2
7	1006	1	2	3	76000				
8	1007	1	3	4	65000				
9	1008	2	3	3	74000				
10	1009	2	3	4	28000				
11	1010	1	2	4	34500				
12	1011	2	3	2	37500				
13									

I3		{=FREQUENCY(E2:E12,H3:H6)}						
	E	F	G	H	I	J	K	L
1	所得							
2	36000		所得分組		次數	百分比		
3	52000		0~	30000	3	27.3%		
4	64000		30000~	50000	3	27.3%		
5	18000		50000~	70000	3	27.3%		
6	22000		70000~		2	18.2%		
7	76000		合計		11	100.0%		
8	65000							
9	74000							
10	28000							
11	34500							
12	37500							
13								

分組資料之次數分配

- 將成績分為~60、60~70、70~80與80~等四組，並求各組之人數及百分比

	A	B	C	D	E	F	G	H
1	學號	成績						
2	8758001	65						
3	8758002	87						
4	8758003	75						
5	8758004	58						
6	8758005	69						
7	8758006	72						
8	8758007	84						
9	8758008	90						
10	8758009	65						
11	8758010	74						
12	8758011	83						
13	8758012	65						
14	8758013	67						
15	8758014	54						
16	8758015	81						
17								

成績	次數	百分比
0~	60	2 13.3%
60~	70	5 33.3%
70~	80	3 20.0%
80~		5 33.3%
合計	15	100.0%

計算累計人數及百分比

- 統計實務上

1. 於求得次數分配表後
2. 再計算累計人數及百分比
3. 求算中位數、四分位數、...
等

- **【累計】** 工作表

[illegible]

- 其計算方法，如J2:J4之累計次數的公式分別為：

- $J_2 : = H_2$
- $J_3 : = H_2 + H_3$
- $J_4 : = H_3 + H_4$


J4				f _x =J3+H4											
A	B	C	D	E	F	G	H	I	J	K	L				
問卷編號	性別	第一題	第二題	所得		第一題	次數	百分比	累計次數	累計%					
1001	1	3	1	36000		1	1	9.1%	1	9.1%					
1002	2	2	2	52000		2	4	36.4%	5	45.5%					
1003	1	1	4	64000		3	6	54.5%	11	100.0%					
1004	2	2	3	18000		合計	11	100.0%	-	-					
1005	1	3	2	22000											
1006	1	2	3	76000											
1007	1	3	4	65000											
1008	2	3	3	74000		所得分組	次數	百分比	累計次數	累計%					
1009	2	3	4	28000		0~ 30000	3	27.3%	3	27.3%					
1010	1	2	4	34500		30000~ 50000	3	27.3%	6	54.5%					
1011	2	3	2	37500		50000~ 70000	3	27.3%	9	81.8%					
						70000~	2	18.2%	11	100.0%					
						合計	11	100.0%	-	-					

- 算得累計次數後
- 選取J2:J4，向右抄給K2:K4即可得到累計百分比
- 其公式分別為
 - K2：=I2
 - K3：=I2+I3
 - K4：=I3+I4
 - 將其安排為百分比格式

[illegible]



以『直方圖』求次數分配並繪圖

- 設定輸出範圍(目前工作表之G2位置)
- 選「累積百分率(M)」，可計算出累計百分比
- 選「圖表輸出(C)」，依原答案順序繪製直方圖



以『直方圖』求次數分配並繪圖

- 按【確定】鈕
- 獲致次數分配表及其直方圖
- 將G6之『其他』，改最後一組
- 答案（4），才符合答案內容



眾數

- **MODE(數值1,數值2, ...)**
- 求一陣列或範圍資料中的眾數（係指在一群體中出現次數最多的那個數值）
- 數值1,數值2,...
- 眾數之儲存格或範圍引數，最多可達255個
- 如果資料組中不包含重複的資料點，本函數將傳回#N/A的錯誤值。眾數、本位數與平均數，均是用來衡量母體的集中趨勢
- **【眾數】** 工作表

	A1		f _x	=MODE(3,2,1,3,1,3,3,2,3)		
	A	B	C	D	E	F
1	3					
2						

眾數

- 類別性（非連續性之名目變數）的資料，
- 如：性別、使用品牌、購買原因、支持那位總統候選人、...等
- 問卷上填答之1、2、3...答案間，並無大小或比例之關係（3並不大於1；3並非1的三倍），只是一個代表類別的數字而已
- 對於這種性質之資料，就以眾數來代表母體的集中趨勢。
 - 如：全班以男性居多、市場上主要以使用A品牌者居多、○○候選人最受選民支持、...
- 這類資料是不會以平均數代表其集中趨勢，像假定求得『使用品牌』欄之均數為2.2，又將如何解釋其意義呢？

C15			fx =MODE(B4:B13)			
	A	B	C	D	E	F
3	問卷編號	使用品牌	性別			
4	1001	3	1	← 1表男性，2表女性		
5	1002	3	2			
6	1003	1	2			
7	1004	2	1			
8	1005	3	2			
9	1006	1	2			
10	1007	3	2			
11	1008	3	1			
12	1009	2	2			
13	1010	1	2			
14						
15	使用者最多之品牌		3	← =MODE(B4:B13)		
16	受訪者主要性別		2	← =MODE(C4:C13)		
17						

中位數

- **MEDIAN(數值1,數值2, ...)**
- 求一陣列或範圍資料的中位數，即將所有數字依大小順序排列後，排列在最中間之數字，其上與其下的數字個數各佔總數的二分之一。如果一組數字為偶數個數，將計算中間兩個數字的平均值。
- 數值1,數值2,...
- 中位數之儲存格或範圍引數，最多可達**255**個

中位數

- 眾數、中位數與平均數，均是用來衡量母體的集中趨勢。
- 通常，對於排順位（次序）之等級資料（如：1表最喜歡、2次之、...，那只表示1將排於2之前的一種順序而已，並無2是1的兩倍之數字關係），係以中位數來當其代表值。

A1		=MEDIAN(10,3,4,5,8,12,7)				
	A	B	C	D	E	F
1	7	← =MEDIAN(10,3,4,5,8,12,7)				
2	6	← =MEDIAN(3,4,5,8,12,7)				取(5+7)/2
3						

中位數

- 資料表==>：
- 對價格的注重順序是優先於對外觀及品牌
- 而對外觀的注重順序又優先於對品牌的注重順序

	B17		fx =MEDIAN(B5:B14)			
	A	B	C	D	E	F
4	問卷編號	對價格的注重順序	對外觀的注重順序	對品質的注重順序		
5	1001	1	2	3	← 1表最注重	
6	1002	1	2	3		
7	1003	1	3	2		
8	1004	3	2	1		
9	1005	2	1	3		
10	1006	1	3	2		
11	1007	1	3	2		
12	1008	3	1	2		
13	1009	1	3	2		
14	1010	1	3	2		
15						
16	注重順序之中位數					
17	價格	1	← =MEDIAN(B5:B14)			
18	外觀	2.5	← =MEDIAN(C5:C14)			
19	品質	3	← =MEDIAN(D5:D15)			
20						

四分位數

- **QUARTILE**(陣列,類型)
- 求一個數值陣列或儲存格範圍的第幾個四分位數。即將所有數字依大小順序排列後，排列在**25%**、**50%**與**75%**之數字。
- 如果該位置介於兩數之間，將計算該點左右兩個數字的平均值。
- 陣列是要求得四分位數的數值陣列或儲存格範圍。

四分位數

- QUARTILE(陣列,類型)
- 類型用以指出要傳回的數值：
- 0 表最小值
- 1 表第一個四分位數(25%處)，Q1
- 2 表第二個四分位數(50%處)，即中位數，Q2
- 3 表第三個四分位數(75%處)，Q3
- 4 表最大值


四分位數

- 最大值減最小值就是全距，是一種離中量數，用來表示群體中各數值之分散情形，數字大表母體中之數值高的很高，但低的卻很低
- 全距是表示一群體全部數值的變動範圍。計算方法很簡單且意義顯明，但反應卻不夠靈敏，當極大、極小數值不變，而其它各項數值皆改變時，全距仍不能反應出變化。且全距易受兩極端數值的影響。
- 第三個四分位數Q3減去第一個四分位數Q1後的一半，即四分位差（Q. D.），其意義為：以母群體居中百分之五十的數值所分散之距離的一半為差量，數字小表分配情況的集中程度高。

[illegible]


偏態

- SKEW(數值1,數值2,...)
- 用以傳回一個分配的偏態，指出一個分配以其平均值為中心的不對稱程度。
- 其值有下列三種情況：
 - =0 此分配為對稱分配
 - >0 此分配為右偏分配，分配集中在低數值方面，不對稱的尾端向正值方向延伸
 - <0 此分配為左偏分配，分配集中在高數值方面，不對稱的尾端向負值方向延伸



峰度

- KURT(數值1,數值2,...)
- 用以傳回一個資料組的峰度（kurtosis），峰度值係顯示與常態分配相較時，尖峰集中或平坦分佈的程度。
- 其情況有三：
 - =3 此分配為常態峰
 - >3 此分配為高狹峰，分佈較為尖峰集中
 - <3 此分配為低闊峰，分佈較為平坦
- 【[峰度](#)】工作表



去除極端值後求均數

- TRIMMEAN(數列或範圍,百分比)
- 可去除指定之百分比的極端值後，再求某數列或範圍之數值的均數。
- 以百分比計算要消除幾個極端值時，會將數值向下取至最接近之2的倍數，以使上下各能排除同樣個數之極端值。如，於**255**個數字中，要排除**0.1**之極端值，應為**3**個數字。但因無法於最小及最大值中各排除**1.5**個數字，故將其捨位為各僅排除**1**個極端數字

D4		fx		=TRIMMEAN(A1:F1,0.4)			
	A	B	C	D	E	F	G
1	10	78	80	83	86	99	
2							
3	原均數	72.66667	← =AVERAGE(A1:F1)				
4	排除最大及最小後之均數		81.75	← =TRIMMEAN(A1:F1,0.4)			
5			81.75	← =AVERAGE(B1:E1)			
6							

第K大的資料

- LARGE(數列或範圍,第幾大)
- 傳回某數列或範圍中，第幾大之數值。

B4		fx =LARGE(\$A\$1:\$E\$1,A4)				
	A	B	C	D	E	F
1	85	90	75	92	68	
2						
3	第幾大					
4	1	92	← =LARGE(\$A\$1:\$E\$1,A4)			
5	2	90	← =LARGE(\$A\$1:\$E\$1,A5)			
6	3	85	← =LARGE(\$A\$1:\$E\$1,A6)			
7	4	75	← =LARGE(\$A\$1:\$E\$1,A7)			
8						



第K小的資料

- SMALL(數列或範圍,第幾小)
- 傳回某數列或範圍中，第幾小之數值。

B4		=SMALL(\$A\$1:\$E\$1,A4)				
	A	B	C	D	E	F
1	85	90	75	92	68	
2						
3	第幾小					
4	1	68	← =SMALL(\$A\$1:\$E\$1,A4)			
5	2	75	← =SMALL(\$A\$1:\$E\$1,A5)			
6	3	85	← =SMALL(\$A\$1:\$E\$1,A6)			
7	4	90	← =SMALL(\$A\$1:\$E\$1,A7)			
8						

敘述統計

- 執行「資料/資料分析...」/選「敘述統計」，按【確定】鈕
- 於『輸入範圍』處，設定要處理之資料範圍（B1:B16）
- 於『分組方式』選「循環」
- 點選「類別軸標記是在第一列上(L)」（因資料含『分數』之字串標記）
- 設定輸出範圍，如安排於目前工作表之D1位置
- 點選「摘要統計(S)」
- 設定要求第2大之數值
- 設定要求第2小之數值
- 按【確】鈕，即可獲致詳細之相關統計數字



敘述統計

- 『敘述統計』也可適用於多組資料

敘述統計

輸入
輸入範圍(I):

分組方式:
☒ 逐欄(C)
☐ 逐列(R)

☒ 類別軸標記是在第一列上(L)

輸出選項
☒ 輸出範圍(Q):

☐ 新工作表(P):

☐ 新活頁簿(W)

☒ 摘要統計(S)

☒ 平均數信賴度(N): %

☒ 第 K 個最大值(A):

☒ 第 K 個最小值(M):

	A	B	C	D	E
1	學號	分數		分數	
2	10279001	86			
3	10279002	45	平均數	76.74	
4	10279003	89	標準誤	2.978674	
5	10279004	76	中間值	82.5	
6	10279005	61	眾數	72	
7	10279006	38	標準差	21.0624	
8	10279007	85	變異數	443.6249	
9	10279008	78	峰度	1.539171	
10	10279009	73	偏態	-1.45421	
11	10279010	90	範圍	85	
12	10279011	26	最小值	15	
13	10279012	83	最大值	100	
14	10279013	82	總和	3837	
15	10279014	83	個數	50	
16	10279015	15	第 K 個最大值(2)	100	
17	10279016	82	第 K 個最小值(2)	22	
18	10279017	72	信賴度(95.0%)	5.985869	
19	10279018	95			
20	10279019	70			

	A	B	C	D	E	F	G
1	學號	期中	期末			期中	期末
2	10279001	86	85				
3	10279002	45	80		平均數	66.33333	69.46667
4	10279003	89	45		標準誤	5.817598	4.297138
5	10279004	76	85		中間值	80	76
6	10279005	61	52		眾數	80	85
7	10279006	38	88		標準差	22.53146	16.64274
8	10279007	85	38		變異數	507.6667	276.981
9	10279008	78	68		峰度	-1.09678	-0.90544
10	10279009	73	85		偏態	-0.67596	-0.69019
11	10279010	90	76		範圍	65	50
12	10279011	26	48		最小值	25	38
13	10279012	83	65		最大值	90	88
14	10279013	82	77		總和	995	1042
15	10279014	83	66		個數	15	15
16	10279015	15	84		第 K 個最大值(2)	88	85
17	10279016	82	61		第 K 個最小值(2)	32	45

小計

- SUBTOTAL(函數類別, 數列或範圍1, 數列或範圍2,...)
- 依指定之函數類別，傳回幾個數列或範圍之某一統計量。式中，斜體字或未加粗之字體，表該部份可省略。
- 其函數類別之代碼及作用分別為：
 - 1 AVERAGE
 - 2 COUNT
 - 3 COUNTA
 - 4 MAX
 - 5 MIN
 - 6 PRODUCT
 - 7 STDEV
 - 8 STDEVP
 - 9 SUM
 - 10 VAR、VARP

	C4		fx =SUBTOTAL(A4,\$A\$1:\$F\$1)			
	A	B	C	D	E	F
1	85	78	69	72	88	90
2						
3	函數類別	作用				
4	1	AVERAGE	80.3			
5	2	COUNT	6.0			
6	3	COUNTA	6.0			
7	4	MAX	90.0			
8	5	MIN	69.0			
9	6	PRODUCT	2.61E+11			
10	7	STDEV	8.7			
11	8	STDEVP	7.9			
12	9	SUM	482.0			
13	10	VAR	75.5			
14	11	VARP	62.9			
15						